Includes easy-entry keystroke listings.

HEWLETT-PACKARD

HP-75

USERS' LIBRARY SOLUTIONS

Finance

NOTICE

TABLE OF CONTENTS

TABLE OF CONTENTS (continued)

PROGRAM DESCRIPTION

BREAKEVEN ANALYSIS

Breakeven analysis is basically a technique for analyzing the relationships among fixed costs, variable costs, and income. Until the breakeven point is reached at the intersection of the total income and total cost lines, the producer operates at a loss. After the breakeven point, each unit produced and sold makes a profit. Breakeven analysis may be represented as follows:



The variables are: fixed costs (F), sales price per unit (P), variable costs per unit (V), number of units sold (U), and gross profit (GP). One can readily evaluate GP or U or P given the other four variables. To calculate the break-even volume, simply let the gross profit equal zero and calculate the number of units sold (U).

To calculate the breakeven volume:

1. Key in the fixed costs.
2. Key in the sales price.
3. Key in the variable costs.
4. Enter the profits as zero.

The program will now compute the number of units, and when you review the data, the answer will be included as part of the items displayed.

To calculate the gross profit at a given volume:

1. Key in the fixed costs.
2. Key in the sales price.
3. Key in the variable costs.
4. Key in the number of units sold.

PROGRAM DESCRIPTION

BREAKEVEN ANALYSIS (continued)

The sales volume may be computed by:

1. Key in the fixed costs.
2. Key in the sales price.
3. Key in the variable costs.
4. Key in the profits desired.

To calculate the required sales price to achieve a given gross profit at a specified sales volume:

1. Key in the fixed costs.
2. Key in gross profits desired.
3. Key in the specified sales volume.
4. Key in the variable costs.

The program will compute the sales price required to achieve the specified gross profits at the chosen sales volume.

Operating Leverage

The degree of operating leverage (OL) at a point is defined as the ratio of the percentage change in net operating income to the percentage change in units sold. The greatest degree of operating leverage is found near the breakeven point where a small change in sales may produce a very large increase in profits. Likewise, firms with a small degree of operating leverage are operating farther from the breakeven point, and they are relatively insensitive to changes in sales volume.

The necessary inputs to calculate the degree of operating leverage are fixed costs (F), sales price per unit (P), variable costs per unit (V), and number of units (U).

The program uses the breakeven routine to enter the data, and to compute the variables needed for the operating leverage.

SAMPLE PROBLEM

Jon Hirsh has a new gadget that he is interested in selling. He wants to know the number of units he needs to sell to break even (profits = zero) with the following facts known:

fixed costs :	$20,000.00
sales price :	15.00
variable costs :	9.81
profits :	0.00

Given this information, what is his leverage at 4000 units?

(Answer : 3854 units)   (Answer : leverage = 27.32)

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	$$ BREAKEVEN ANALYSIS $$	
1a	Select Breakeven analysis		
	function	PRESS Breakeven OR Leverage	B [RTN]
2	Enter fixed costs figure	Fixed Costs =	20000 [RTN]
3	Enter sales price figure	Sales Price =	15 [RTN]
4	Enter variable costs	Variable Costs =	9.81 [RTN]
5	Skip number of units for now	Number of Units =	[RTN]
6	Enter breakeven profits	Profits =	0 [RTN]
7	View answers:		
	Fixed costs:	Fixed Costs 20000.00	[RTN]
	Sales price:	Sales Price 15.00	[RTN]
	Variable costs:	Variable Costs 9.81	[RTN]
	Number of units:	Number of Units 3853.56	[RTN]
	Profits:	Profits 0	[RTN]
8	Rerun for leverage computation	Run again, View again, or End?	R [RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
9	Perform leverage analysis	PRESS Breakeven OR Leverage	L [RTN]
10	Enter the 4000 units sold	Number of Units sold	4000 [RTN]
11	View the parameters and the		
	answers next:		
	Fixed costs:	Fixed Costs 20000.00	[RTN]
	Sales price:	Sales Price 15.00	[RTN]
	Variable costs:	Variable Costs 9.81	[RTN]
	Number of units:	Number of Units 4000.00	[RTN]
	Profits	Profits 0.00	[RTN]
	Leverage:	Leverage 27.32	[RTN]
12	End the program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	$$ BREAKEVEN ANALYSIS $$	
1a	Select Breakeven or Leverage	PRESS Breakeven OR Leverage	B or L [RTN]
	(B = Breakeven selected)		
2	These five prompts will	Fixed Costs =	V(1) [RTN]
	loop through until four	Sales Price =	V(2) [RTN]
	are answered with a value.	Variable Costs =	V(3) [RTN]
	[RTN] skips to next prompt.	Number of Units =	V(4) [RTN]
		Profits =	V(5) [RTN]
3	After four entries, the	Fixed Costs V(1)	[RTN]
	answers are displayed.	Sales Price V(2)	[RTN]/[BACK]
	[RTN] steps to the next answer,	Variable Costs V(3)	[RTN]/[BACK]
	[BACK] steps back to the prior	Number of Units V(4)	[RTN]/[BACK]
	answer.	Profits V(5)	[RTN]/[BACK]
4	Program options:	Run again, View again, or End?	R, V, or E [RTN]
	If 'V' then 3. If 'R' then 1a.		
	If 'E' then 8.		
	(L = Leverage selected)		
5	Enter number of units sold	Number of units sold	V(4) [RTN]
6	The answers are displayed	Fixed Costs V(1)	[RTN]
	[RTN] steps to next answer,	Sales Price V(2)	[RTN]/[BACK]
	[BACK] steps back to prior	Variable Costs V(3)	[RTN]/[BACK]
	answer.	Number of Units V(4)	[RTN]/[BACK]
		Profits V(5)	[RTN]/[BACK]
		Leverage V(6)	[RTN]/[BACK]
7	Goto step 4		
8	End of program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
V(1)	Fixed costs	C	Input counter (0≤C≤4)
V(2)	Sales price	N	Formula pointer
V(3)	Variable costs	X	Loop index
V(4)	Number of units sold	A$	Output labels
V(5)	Profits	K$	Single key response
V(6)	Leverage	Q$	User interaction
		X$	User interaction

NOTES AND REFERENCES

Note: When viewing either breakeven analysis of leverage answers, [RTN] goes to next answer, [BACK] steps to prior answer.

Reference: Breakeven Analysis, HP-12C Solutions Handbook, p. 54.

PROGRAM LISTING

```
 10 ! Breakeven analysis
 20 ! Computes units sold,
 30 ! variable costs,fixed costs,
 40 ! selling price, or profits
 50 ! given the other four.
 60 !
 70 !  Given fixed costs,
 80 ! sales price, variable
 90 ! costs, and the number of
100 ! units sold, it will
110 ! compute the operating
120 ! leverage.
130 !
140 ! Revision 11/01/82
150 !
160 DELAY 2
170 DISP "    $$ Breakeven analysis $$"     -Sign on message
180 DIM A$[96]
190 IMAGE 16a,2x,7d.dd
200 IMAGE 7d
210 FOR X=1 TO 6 @ V(X)=0 @ NEXT X
220 !
230 ! Single, uppercase key in              -Returns single uppercase
                                             character
240 DEF FNK$
250 K$=KEY$ @ IF K$="" THEN 250
260 FNK$=UPRC$(K$)
270 END DEF
280 !
290 ! Prompt and label strings              -Initialize prompts
300 A$[1,16]="Fixed Costs     "
310 A$[17,32]="Sales Price     "
320 A$[33,48]="Variable Costs  "
330 A$[49,64]="Number of Units "
340 A$[65,80]="Profits         "
350 A$[81,96]="Leverage        "
360 !
370 ! Breakeven or Leverage                 -Select program option
380 DISP "PRESS ";CHR$(194);"reakeven O
    R ";CHR$(204);"everage ";
390 INPUT "";K$
400 IF UPRC$(K$)="B" THEN 630
410 IF UPRC$(K$)#"L" THEN 380
420 ! Fall into leverage routine
430 INPUT "Number of units sold "; X$ @    -Leverage routine
     IF X$="" THEN 430
440 IF X$="Q" THEN 1170                     -If user enters 'Q' then quit
450 ON ERROR GOTO 470
460 V(4)=VAL(X$) @ OFF ERROR @ GOTO 480
470 DISP "Oops...";  @ GOTO 430
480 IF V(4)*(V(2)-V(3))#V(1) THEN 510      -Test for infinite leverage
490 DISP "Leverage is infinite at ";V(4
    )
500 GOTO 1110
```

PROGRAM LISTING

```
510 V(6)=V(4)*(V(2)-V(3))/(V(4)*(V(2)-V      -Compute leverage
    (3))-V(1))
520 ! display leverage results
530 FOR X=1 TO 6
540 DISP USING 190 ; A$[X*16-15,X*16]&"      -Show label and quantity
     = ";V(X)
550 Q$=FNK$
560 IF NUM(Q$)#8 AND NUM(Q$)#13 THEN 55      -Wait for 'RTN' or 'BACK' keys
    0
570 IF NUM(Q$)#8 THEN 590
580 X=X-2 @ IF X<0 THEN X=0                  -Decrement counters for 'BACK'
                                              key
590 NEXT X
600 GOTO 1110                                -Proceed to options menu
610 ! Data entry loop
620 ! Set C and clear variables
630 C=4 @ FOR X=1 TO 5 @ V(X)=0 @ NEXT
    X
640 ! Disp
650 FOR X=1 TO 5
660 DISP A$[X*16-15,X*16];" = ";             -Show next label for input
670 IF V(X) THEN DISP USING 200 ; V(X)       -Show previous data that has
                                              been entered
680 INPUT ""; X$ @ IF X$="" THEN 740
690 IF X$="Q" THEN 1170
700 ON ERROR GOTO 720
710 V(X)=VAL(X$) @ C=C-1 @ OFF ERROR @
    GOTO 730
720 DISP "Oops....";  @ GOTO 660
730 IF NOT C THEN X=5
740 NEXT X
750 IF C THEN 640
760 ! Compute answers
770 GOSUB 870
780 ! display the results
790 GOSUB 1030
800 !
810 ! Loop back for more inputs
820 GOTO 610
830 !
840 ! Subroutines follow
850 !
860 ! Find proper formula
870 N=0 @ FOR X=1 TO 5
880 IF V(X)=0 THEN N=X @ GOTO 900
890 NEXT X
900 ON N GOSUB 930,950,970,990,1010
910 RETURN
920 !
930 V(1)=V(4)*(V(2)-V(3))-V(5) @ RETURN
940 !
950 V(2)=V(3)+(V(5)+V(1))/V(4) @ RETURN
960 !
970 V(3)=V(2)-(V(5)+V(1))/V(4) @ RETURN
```

PROGRAM LISTING

```
 980 !
 990 V(4)=V(5)+V(1)/(V(2)-V(3)) @ RETURN
1000 !
1010 V(5)=V(4)*(V(2)-V(3))-V(1) @ RETURN
1020 !
1030 ! disp results
1040 FOR X=1 TO 5
1050 DISP USING 190 ; A$[X*16-15,X*16]&"
      = ";V(X)
1060 Q$=FNK$
1070 IF NUM(Q$)#8 AND NUM(Q$)#13 THEN 10
     60
1080 IF NUM(Q$)#8 THEN 1100
1090 X=X-2 @ IF X<0 THEN X=0
1100 NEXT X
1110 GOSUB 1140
1120 IF Q$="V" THEN 1040
1130 IF Q$="R" THEN 380 ELSE 1170
1140 DISP CHR$(210);"un again, ";CHR$(21
     4);"iew again, or ";CHR$(197);
1150 INPUT "nd ","R"; Q$ @ Q$=UPRC$(Q$)
1160 IF Q$="V" OR Q$="R" OR Q$="E" THEN
     RETURN
1170 DELAY 1 @ DISP @ STOP
```

PROGRAM DESCRIPTION

SECURITIES EARNINGS

Given the expected growth rate, current price per share, earnings per share, initial growth rate, years of declining growth rate, the discount rate, current payment and final P/E ratio, compute the number of years of constant growth to justify the current price of the stock.

If an HP-IL printer is attached to the system, and defined by a "printer is" command, this program will print the results.

SAMPLE PROBLEM

Given a stock with a share price of $66 and a growth rate of 4%, compute the number of years to justify the current price. The earnings per share are 2.87; the initial growth rate is 10%. The discount rate is 12%, the current payment ratio is 48%, the number of years of declining growth are 6 and the final P/E ratio should be 12.

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	Sign-on message	$ Securities Earnings $	
2	Enter data	Growth rate (decimal)?	0.04 [RTN]
		Current share price?	66 [RTN]
		Earnings per share?	2.87 [RTN]
		Initial growth rate in EPS?	0.10 [RTN]
		Years of declining growth?	6 [RTN]
		Discount rate (decimal)?	0.12 [RTN]
		Current payout ratio (decimal)?	0.48 [RTN]
3	Change computed P/E ratio	Your PE ratio is 7.5	[RTN]
		If this is not satisfactory	[RTN]
		Enter the new PE ratio,	[RTN]
		Otherwise, enter zero?	12 [RTN]
4	Perform computations	>>> Calculating <<<	
5	Display results	Current share price 66	[RTN]
		Earnings per share 2.87	[RTN]
		Initial growth rate (EPS) .1	[RTN]
		Final growth rate (EPS) .04	[RTN]
		Yrs of declining growth 6	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Discount rate .12	[RTN]
		Current payout ratio .48	[RTN]
		Final PE ratio 12	[RTN]
		Price 66 assumes 41 years	[RTN]
		Present value 59.18	[RTN]
		Intrinsic value for N1=40 is	[RTN]
		58.75	
		Price in 46 years 2337.91	[RTN]
6	End program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	Sign-on message	$ Securities Earnings $	
2	Enter data	Growth rate (decimal)?	n [RTN]
		Current share price	P2 [RTN]
		Earnings per share?	E [RTN]
		Initial growth rate in EPS?	G3 [RTN]
		Years of declining growth?	N2 [RTN]
		Discount rate (decimal)?	K [RTN]
		Current payout ratio (decimal)?	PO [RTN]
3	User has option to change the	Your PE ratio is I3	[RTN]
	computed PE ratio	If this is not satisfactory,	[RTN]
		Enter the new PE ratio,	[RTN]
		Otherwise enter zero?	I3 [RTN]
4	Perform computations	<<< Calculating >>>	
5	Display results	Current share price P2	[RTN]/[BACK]
	[RTN] advances to next display	Earnings per share E	[RTN]/[BACK]
	[BACK] displays prior entry	Initial growth rate (EPS) G3	[RTN]/[BACK]
	[TAB] ends program	Final growth rate (EPS)	[RTN]/[BACK]
	If an HP-IL printer is attached	Yrs of declining growth N2	[RTN]/[BACK]
	and if the user defined the	Discount rate K	[RTN]/[BACK]
	printer by "printer is" then	Current payout ratio PO	[RTN]/[BACK]
	this will be printed out one	Final PE ratio I3	[RTN]/[BACK]
	line at a time	Price P2 assumes N1 years	[RTN]/[BACK]
		Present value S	[RTN]/[BACK]
		Intrinsic value for N1=N1 is X	[RTN]/[BACK]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Price in X years X	[RTN]/[BACK]
	Program options	Run again, View again, or End?	R,V, or E [RTN]
	If 'R' is pressed, goto 2		
	If 'V' is pressed, goto 5		
	If 'E' is pressed stop		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
D(*)	Dividends	P(*)	Holding values
E(*)	Earnings per share	P	Share price
F2	Average growth rate	PO	Payout ratio
F3	Weight in present value routine	P1	Temporary value
G1	Growth rate in EPS	P2	Share price
G2	Temporary growth rate	P9	Payout ratio
G3	Initial growth rate	QO	Annual change in payout ratio
I	Index value	S	Present value
I3	P/E ratio	SO	Intrinsic value
K	Discount rate	X9	Parameter in rounding function
N	Sum of constant & declining growth yrs	Y9	Parameter in rounding function
N1	Years of constant growth	(*)	Indicates subscript variable
N2	Years of declining growth	Q$	Keyboard response
		X$	Alpha value input convert-ed to numeric value

PROGRAM LISTING

```
 10 ! Find the number of years
 20 ! of constant growth in the
 30 ! earning per share (EPS) to
 40 ! justify the current share
 50 ! price.
 60 !
 70 !
 80 ! revision 11/01/82
 90 !
100 DELAY 2 @ DISP "      $ Securities E    -Initialize delay and display
    arnings $"                               sign-on message
110 DIM D(60),E(60),P(8)
120 P9=.6 @ G3=.04                          -Initialize default values for
                                             payout ratio, growth rate
130 !
140 ! Round X9 to P9 decimal digits         -FNR rounds X9 to P9 places
150 !
160 DEF FNR(X9,Y9) = INT(X9*10^Y9+5/10^
    Y9)/10^Y9
170 !
180 INPUT "Growth rate?"; X$ @ ON ERROR     -Input for growth rate; setup
     GOTO 200                                error trap
190 G3=VAL(X$) @ OFF ERROR @ GOTO 210
200 GOSUB 1170 @ GOTO 180                   -On error display error message
                                             and ask again
210 P(4)=G3
220 INPUT "Current share price?"; X$ @
    ON ERROR GOTO 240
230 P=VAL(X$) @ OFF ERROR @ GOTO 250
240 GOSUB 1170 @ GOTO 220
250 P(1)=P
260 INPUT "Earnings per share?"; X$ @ O
    N ERROR GOTO 280
270 E(1)=VAL(X$) @ OFF ERROR @ GOTO 290
280 GOSUB 1170 @ GOTO 260
290 P(2)=E(1)
300 INPUT "Initial growth rate in EPS?"
    ; X$ @ ON ERROR GOTO 320
310 G1=VAL(X$) @ OFF ERROR @ GOTO 330
320 GOSUB 1170 @ GOTO 300
330 P(3)=G1
340 INPUT "Years of declining growth?";
     X$ @ ON ERROR GOTO 360
350 N2=VAL(X$) @ OFF ERROR @ GOTO 370
360 GOSUB 1170 @ GOTO 340
370 P(5)=N2
380 INPUT "Discount rate (decimal)?"; X
    $ @ ON ERROR GOTO 400
390 K=VAL(X$) @ OFF ERROR @ GOTO 410
400 GOSUB 1170 @ GOTO 380
410 P(6)=K
420 INPUT "Current payout ratio?"; X$ @
     ON ERROR GOTO 440
430 P0=VAL(X$) @ OFF ERROR @ GOTO 450
```

PROGRAM LISTING

```
440 GOSUB 1170 @ GOTO 420
450 P(7)=P0
460 E(1)=E(1)*(1+G1)
470 D(1)=P0*E(1)
480 DISP "Your PE ratio is ";P9/(K-G3)      -Display computed P/E ratio,
    @ GOSUB 1220 @ IF NUM(Q$)=8 THEN 48       wait for keyboard
    0
490 DISP "If this is not satisfactory,"
     @ GOSUB 1220 @ IF NUM(Q$)=8 THEN G
    OTO 480
500 DISP "enter the new PE ratio." @ GO
    SUB 1220 @ IF NUM(Q$)=8 THEN GOTO 4
    90
510 INPUT "Otherwise, enter zero?"; X$
    @ ON ERROR GOTO 530
520 I3=VAL(X$) @ OFF ERROR @ GOTO 540
530 GOSUB 1170 @ GOTO 510
540 IF I3#0 THEN P(8)=I3 ELSE P(8)=P9/(      -Use new value for computing
    K-G3)                                      new P/E ratio if valid.
550 IF K>G3 THEN 580
560 DISP "Discount must be > ";G3 @ GOS      -If discount < growth rate,
    UB 1220 @ IF NUM(Q$)=8 THEN GOTO 56        quit.
    0
570 GOTO 1280
580 S0=E(1)*P9/(K-G3)                        -Compute the intrinsic value
590 DISP "   >>>>>> Calculating <<<<<<"      -Indicate that calculation is
                                               in progress
600 S=0
610 FOR N1=1 TO 40                           -Begin computations for each
                                               year of constant growth
620 S0=S
630 Q0=P9-P0
640 N=N1+N2                                  -N is the total number of years
                                               involved
650 IF N<=5 THEN 670                         -If total number of years < 5,
                                               skip payout ratio
660 Q0=(P9-P0)/(N-5)
670 P1=P0
680 IF N1=1 THEN 750
690 FOR I=2 TO N1                            -Compute dividends and earnings
                                               per share for each year
700 E(I)=E(I-1)*(1+G1)
710 D(I)=E(I)*P1
720 IF I<5 THEN 740
730 P1=P1+Q0
740 NEXT I
750 G2=G1                                    -Setup temporary growth rate in
                                               EPS
760 F2=(G1-G3)/(N2+1)
770 FOR I=N1+1 TO N
780 G2=G2-F2
790 E(I)=E(I-1)*(1+G2)
800 D(I)=E(I)*P1
810 IF I<5 THEN 830
```

PROGRAM LISTING

```
820 P1=P1+Q0
830 NEXT I
840 D(N)=P9*E(N)                          -Compute dividends for final
                                           year of analysis
850 P2=D(N)/(K-G3)
860 IF I3=0 THEN 880
870 P2=I3*E(N)
880 P2=P2/(1+K)^N                         -Compute share price in final
                                           year
890 S=0
900 F3=1
910 FOR I=1 TO N                          -End of computation loop
920 F3=F3/(1+K)
930 S=S+D(I)*F3
940 NEXT I
950 S=S+P2
960 IF P<S THEN 980
970 NEXT N1
980 PRINT "Current share price ";FNR(P(   -Display results
    1),2) @ GOSUB 1220 @ IF NUM(Q$)=8 T
    HEN 980
990 PRINT "Earnings per share ";FNR(P(2
    ),2) @ GOSUB 1220 @ IF NUM(Q$)=8 TH
    EN GOTO 980
1000 PRINT "Initial growth rate (EPS) ";
     FNR(P(3),1) @ GOSUB 1220 @ IF NUM(Q
     $)=8 THEN GOTO 990
1010 PRINT "Final growth rate (EPS) ";FN
     R(P(4),2) @ GOSUB 1220 @ IF NUM(Q$)
     =8 THEN GOTO 1000
1020 PRINT "Yrs of declining growth ";FN
     R(P(5),1) @ GOSUB 1220 @ IF NUM(Q$)
     =8 THEN GOTO 1010
1030 PRINT "Discount rate ";FNR(P(6),2)
     @ GOSUB 1220 @ IF NUM(Q$)=8 THEN 10
     20
1040 PRINT "Current payout ratio ";FNR(P
     (7),2) @ GOSUB 1220 @ IF NUM(Q$)=8
     THEN GOTO 1030
1050 PRINT "Final P/E ratio ";FNR(P(8),2
     ) @ GOSUB 1220 @ IF NUM(Q$)=8 THEN
     GOTO 1040
1060 PRINT "Price ";FNR(P,2);" assumes "
     ;N1;" years" @ GOSUB 1220 @ IF NUM(
     Q$)=8 THEN 1050
1070 PRINT "Present value ";FNR(S,2) @ G
     OSUB 1220 @ IF NUM(Q$)=8 THEN 1060
1080 PRINT "Intrinsic value for N1=";N1-
     1;" is " @ GOSUB 1220 @ IF NUM(Q$)=
     8 THEN GOTO 1070
1090 PRINT FNR(S0,2) @ GOSUB 1220 @ IF N
     UM(Q$)=8 THEN GOTO 1080
1100 PRINT "Price in ";N;"years is ";FNR
     (P2*(1+K)^N,2) @ GOSUB 1220
```

PROGRAM LISTING

```
1110 IF NUM(Q$)=8 THEN 1080
1120 GOTO 1280
1130 !
1140 ! Prepend error message to
1150 ! input prompt                        -When input error occurs, show
                                           error and ask again
1160 !
1170 DISP "Oops...";                      -Error message for input
1180 RETURN
1190 !
1200 ! monitor the keyboard               -Monitor the keyboard
1210 !
1220 Q$=KEY$ @ IF NUM(Q$)#8 AND NUM(Q$)#  -Accept only 'RTN', 'BACK' or
     13 AND NUM(Q$)#142 THEN 1220          'TAB' keys
1230 Q$=UPRC$(Q$)
1240 RETURN
1250 !
1260 ! Options menu
1270 !
1280 DISP CHR$(210);"un again,";CHR$(214  -Display options menu
     );"iew again, or ";CHR$(197); @ INP
     UT "nd ";Q$
1290 Q$=UPRC$(Q$)
1300 IF Q$#"E" AND Q$#"R" AND Q$#"V" THE  -Accept only 'R', 'V' or 'E'
     N GOTO 1280
1310 IF Q$="R" THEN 180
1320 IF Q$="V" THEN 980
1330 DELAY 1 @ DISP @ STOP
```

PROGRAM DESCRIPTION

NOTES

This program accepts the face value of a note, its discount rate (as a percentage) and the days to maturity of the note and computes the discount amount and the net cost of the note.

The formula for the discount amount is:

$$D = F*I/100*N/360$$

Where: F = Face value of note
I = Discount rate
N = Days to maturity (360 day calendar)

The net cost of the note is equal to face value less discount amount.

SAMPLE PROBLEM

Bob Johnson is purchasing a $150,000 note that will mature in 126 days. If the discount rate is 14.5%, what is the discount amount, and what is the net cost of the note?

Answer: $7,612.5 and $142,387.5

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run the program		
	Sign-on message	$ Notes $	
2	Enter the face value of the note	Face (future) value?	150000 [RTN]
	Enter the discount rate	Discount rate (%)?	14.5 [RTN]
	Enter the days to maturity	Days to maturity?	126 [RTN]
3	View the results	Face (future) value $150000	[RTN]
		Discount rate 14.5%	[RTN]
		126 days to maturity	[RTN]
		Discount amount is $7612.5	[RTN]
		Net cost is $142387.5	[RTN]
4	Options menu - end program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run the program		
	Sign-on message	$ Notes $	
2	Enter face value of note	Face (future) value?	F [RTN]
	Enter the discount rate	Discount rate (%)?	I [RTN]
	Enter the days to maturity	Day to maturity?	N [RTN]
3	View the results:	Face (future) value F	[RTN]
	[RTN] advances to meet item	Discount rate I (%)	[RTN]/[BACK]
	[BACK] shows the prior item	N days to maturity	[RTN]/[BACK]
	[TAB] terminates the program	Discount amount is D	[RTN]/[BACK]
		Net cost is F-D	[RTN]/[BACK]
4	Program options	Run again, View again, or End?	R,V, or E [RTN]
	If 'R' is pressed, goto 2		
	If 'V' is pressed, goto 3		
	If 'E' is pressed, end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
D	The discount amount	X	Number to be rounded
F	Face value of the note	K$	Keyboard input
I	Discount rate	Q$	Value of the key used
N	Days to maturity	K$	Input value
P	Precision of rounding function		

NOTES AND REFERENCES

Note: The program uses a 360 day calendar.

Reference: Poole, Lon and Mary Borchers, SOME COMMON BASIC PROGRAMS, 2nd ed., (Osborne-McGraw-Hill, 1977), p. 27-28.

PROGRAM LISTING

10 ! Notes:	
20 ! calculates the cost of	
30 ! a note, given face	
40 ! (future) value, the discount	
50 ! rate, and the number of	
60 ! days to maturity.	
70 !	
80 ! revision 11/01/82	
90 !	
100 DELAY 2 @ DISP " $ Notes $"	-Display sign-on message
110 ! round X to P decimal digits	-This function rounds X to P digits
120 !	
130 DEF FNR(X,P)	
140 FNR=INT(X*10^P+5/10^P)/10^P	
150 END DEF	
160 !	
170 ! single upper-case key in	-Returns single uppercase character
180 !	
190 DEF FNK$	
200 K$=KEY$ @ IF K$="" THEN 200	
210 FNK$=UPRC$(K$)	
220 END DEF	
230 ! begin data input	
240 !	
250 INPUT "Face (future) value?"; X$ @ ON ERROR GOTO 270	-Display prompt, set error trap
260 F=VAL(X$) @ OFF ERROR @ IF F<=0 THEN 270 ELSE 280	-Convert alpha input to numeric, test data validity
270 DISP "Oops..."; @ GOTO 250	-Prepend error message to input prompt and ask again
280 INPUT "Discount rate (%)?"; X$ @ ON ERROR GOTO 300	
290 I=VAL(X$) @ OFF ERROR @ IF I<=0 THEN 300 ELSE 310	
300 DISP "Oops..."; @ GOTO 280	
310 INPUT "Days to maturity?"; X$ @ ON ERROR GOTO 330	
320 N=VAL(X$) @ OFF ERROR @ IF N<=0 THEN 330 ELSE 370	
330 DISP "Oops..."; @ GOTO 310	
340 !	
350 ! Compute results	
360 !	
370 D=F*I/100*N/360	-Compute the discount amount
380 !	
390 ! Output values	
400 !	
410 DISP "Face (future) value $";F @ GOSUB 500 @ IF NUM(Q$)=8 THEN 410	-Display the face value, wait for keyboard
420 DISP "Discount rate";I;"%" @ GOSUB 500 @ IF NUM(Q$)=8 THEN 410	-Display next line of output, monitor keyboard
430 DISP N;"days to maturity" @ GOSUB 500 @ IF NUM(Q$)=8 THEN 420	

PROGRAM LISTING

```
440 DISP "Discount amount is $";FNR(D,1
    ) @ GOSUB 500 @ IF NUM(Q$)=8 THEN 4
    30
450 DISP "Net cost is $";FNR(F-D,1) @ G
    OSUB 500 @ IF NUM(Q$)=8 THEN 440
460 GOTO 560
470 !
480 ! Monitor the keyboard
490 !
500 Q$=FNK$ @ IF NUM(Q$)#13 AND NUM(Q$)    -Accept only 'RTN', 'BACK' or
    #8 AND NUM(Q$)#142 THEN 500             'TAB' keys
510 IF NUM(Q$)=142 THEN 610                -Quit if 'TAB' was pressed
520 RETURN
530 !
540 ! Present options menu
550 !
560 DISP CHR$(210);"un again,";CHR$(214    -Display options menu
    );"iew again, or ";CHR$(197); @ INP
    UT "nd?";Q$
570 Q$=UPRC$(Q$)
580 IF Q$#"R" AND Q$#"E" AND Q$#"V" THE    -Accept only the 'R', 'V', or
    N 560                                   'E' keys
590 IF Q$="R" THEN 100
600 IF Q$="V" THEN 410
610 DELAY 1 @ DISP @ STOP
```

PROGRAM DESCRIPTION

BOND PRICE AND YIELD

The program uses a 360 day calendar and given the redemption date, settlement date, annual coupon rate, redemption value, annual yield or bond price, will compute either the bond price or annual yield for semi-annual coupon bonds.

The program computes the number of coupon periods between the settlement and redemption dates, and uses this in computing the bond price.

If an HP-IL printer is attached and defined by "printer is", this program will print the results.

SAMPLE PROBLEM

Mark is interested in purchasing a bond that yields 6.23%. The bond has a coupon of 5%. If the settlement date is 7,7, 1983 and the redemption date is 6,30, 1985, what is the price of the bond?

Jane is buying a 3.22% bond for $89.43. The settlement date is 3,23, 1982 and the maturity date is 5,25, 1988. What is the bond's yield?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	$ Bond Price and Yield $	
2	Enter settlement date:	Settlement date (mm,dd,yyyy)?	7,7,1983 [RTN]
	Enter redemption date:	Redemption date (mm,dd,yyyy)?	6,30,1985 [RTN]
	Enter annual coupon rate:	Annual coupon rate (%)?	5 [RTN]
	Redemption value=100, so skip	Redemption value?	[RTN]
	Enter annual yield (%):	Annual yield (%)?	6.23 [RTN]
3	Display answers:	Number of coupon periods 3.96	[RTN]
		Annual coupon rate (%) 5	[RTN]
		Redemption value 100	[RTN]
		Annual yield 6.23	[RTN]
		Bond price 97.74	[RTN]
4	Program options:	Run again, View again, or End?	R [RTN]
4a	Sign-on message	$ Bond Price and Yield $	
5	Second problem:	Settlement date (mm,dd,yyyy)?	3,23,1982 [RTN]
		Redemption date (mm,dd,yyyy)?	5,25,1988 [RTN]
		Annual coupon rate (%)?	3.22 [RTN]
		Redemption value?	[RTN]
		Annual yield (%)?	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Bond price?	89.43 [RTN]
6	Display answers:	Number of coupon periods 12.34	[RTN]
		Annual coupon rate (%) 3.22	[RTN]
		Redemption value 100	[RTN]
		Annual yield 5.246	[RTN]
		Bond price 89.43	[RTN]
7	End program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program.		
1a	See sign-on message	$ Bond Price and Yield $	
2	Enter settlement date:	Settlement date (mm,dd,yyyy)?	mm,dd,yyyy [RTN]
	Enter redemption date:	Redemption date (mm,dd,yyyy)?	mm,dd,yyyy [RTN]
	Enter coupon rate:	Annual coupon rate (%)?	n [RTN]
	Enter redemption value if it is	Redemption value?	n [RTN] or [RTN]
	not equal to 100		
	Enter annual yield (%):	Annual yield (%)?	n [RTN]
	If annual yield is zero:	Bond price?	n [RTN]
3	Display results - press [RTN]	Number of coupon periods	[RTN]
	to view next item; [BACK]	Annual coupon rate (%)	[RTN]/[BACK]
	to view prior item; [TAB] to	Redemption value	[RTN]/[BACK]
	end program	Annual yield	[RTN]/[BACK]
		Bond price	[RTN]/[BACK]
4	Display options menu	Run again, View again, or End?	R,V,or E [RTN]
	If 'R' is pressed goto 1a		
	If 'V' is pressed goto 3		
	If 'E' is pressed end program.		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A	Parameter in bond price computation	R1	Annual coupon rate - for output
B	Bond price	S1	Parameter in year conversion routine
C	Temporary variable in bond yield	S2	Parameter in year conversion routine
C1	Number of coupon periods	V1	Redemption value
D1	Converted days for settlement date	X	Parameter in the rounding function
D2	Converted days for redemption date	Y	Yield
D3	Difference between D2 & D1	Y1	Temporary variable in yield routine
D8	Day of month - redemption date	Y2	Yield - held for output
D9	Day of month - settlement date	Y8	Year number - redemption
I	Temporary variable in bond yield	Y9	Year number - settlement
J	Temporary variable in bond price	Z	Parameter in rounding function
M8	Month number - redemption	K$	Upper case keyboard input
M9	Month number - settlement	Q$	Keyboard input
P	Computed bond price in yield computations	X$	Alphabetic input values
R	Annual coupon rate - computations	Y$	that are converted to
		Z$	numeric for later use

PROGRAM LISTING

```
10 ! Given the number of coupon
20 ! periods between settlement
30 ! date and redemption date,
40 ! the annual coupon rate as
50 ! a percent, the redemption
60 ! value (if other than 100),
70 ! and the annual yield as a
80 ! percent, compute the 'flat'
90 ! price.
100 !
110 ! revision 11/01/82
120 !
130 DELAY 2 @ DISP "   $ Bond Price and
     Yield $"
140 !
150 ! convert to 360-day calendar       -Function to convert date to
                                         360 day calendar
160 !
170 DEF FNA(S1,S2,S3) = 360*S3+30*S1+S2
180 !
190 ! Compute price of bond            -Compute the price of the bond
200 !
210 DEF FNB(A) = R/2*(1/(Y/2/((1+Y/2)^D
    3-1)+Y/2))+100*(1+Y/2)^(-D3)
220 !
230 ! Round Z to P decimal places      -Round Z to P places
240 !
250 DEF FNR(Z,P) = INT(Z*10^P+5/10^P)/1
    0^P
260 !
270 ! Single upper-case key in         -Monitor keyboard and return
                                         uppercase character
280 !
290 DEF FNK$
300 K$=KEY$ @ IF K$="" THEN 300
310 FNK$=UPRC$(K$)
320 END DEF
330 !
340 ! begin data input
350 !
360 INPUT "Settlement date(mm,dd,yyyy)?  -Accept input values for
    "; X$,Y$,Z$ @ ON ERROR GOTO 380      settlement date
370 M9=VAL(X$) @ D9=VAL(Y$) @ Y9=VAL(Z$  -Convert input to numeric,
    ) @ OFF ERROR @ D1=FNA(M9,D9,Y9) @   convert to 360 day calendar
    GOTO 390
380 GOSUB 1060 @ GOTO 360                -Process error and ask again
390 INPUT "Redemption date (mm,dd,yyyy)
    ?"; X$,Y$,Z$ @ ON ERROR GOTO 410
400 M8=VAL(X$) @ D8=VAL(Y$) @ Y8=VAL(Z$
    ) @ OFF ERROR @ D2=FNA(M8,D8,Y8) @
    GOTO 420
410 GOSUB 1010 @ GOTO 390
420 C1=(D2-D1)/180                       -Compute number of coupon
                                         periods for semi-annual
                                         coupons
```

PROGRAM LISTING

```
430 INPUT "Annual coupon rate (%)?"; X$
     @ ON ERROR GOTO 450
440 R1=VAL(X$) @ R=R1 @ OFF ERROR @ GOT
    O 460
450 GOSUB 1010 @ GOTO 430
460 INPUT "Redemption value?"; X$ @ ON
    ERROR GOTO 490
470 IF X$="" THEN V1=100 @ OFF ERROR @
    GOTO 500
480 V1=VAL(X$) @ OFF ERROR @ GOTO 500
490 GOSUB 1010 @ GOTO 460
500 INPUT "Annual yield (%)?"; X$ @ ON
    ERROR GOTO 530
510 IF X$="" THEN Y=0 @ Y2=Y @ GOTO 550
520 Y=VAL(X$) @ Y2=Y @ OFF ERROR @ GOTO
     540
530 GOSUB 1010 @ GOTO 500
540 IF Y#0 THEN 740
550 INPUT "Bond price?"; X$ @ ON ERROR
    GOTO 570
560 B=VAL(X$) @ OFF ERROR @ GOTO 580
570 GOSUB 1010 @ GOTO 550
580 D3=(FNA(M8,D8,Y8)-FNA(M9,D9,Y9))/36    -Compute the number of years
    0                                       for the life of the bond
590 I=R+(100-B)/D3 @ D3=2*D3               -Interest on the bond
600 C=(B+100)/2 @ Y=I/C                    -Compute the initial guess of
                                            the yield
610 IF Y<=0 THEN DISP "Yield is negativ    -If the yield is negative,
    e or zero" @ Y2=Y*100 @ GOTO 850        inform user and goto end
620 IF FNB(Y)<B THEN Y1=Y/2 @ GOTO 650     -If computed yield generates
    ELSE Y=2*Y @ GOTO 620                   bond price, exit
630 Y=Y-2*Y1
640 Y=Y+Y1
650 P=FNB(Y)                               -Compute test bond price using
                                            estimated yield
660 IF ABS(P-B)<.001 THEN 690              -If the difference is less than
                                            .1 cent, exit
670 Y1=Y1/2
680 IF P-B<0 THEN 630 ELSE 640
690 Y2=Y*100
700 GOTO 850
710 !
720 ! Compute bond price
730 !
740 ! IF Y1=0 THEN 360                       -If input yield is zero, error
                                            exists; ask again
750 J=1-FP(C1)
760 R=R/100 @ Y=Y/100
770 IF C1<=1 THEN 800                      -If number of coupon periods <
                                            1 use different formula
780 B=V1*(1+Y/2)^(-C1)+100*(R/Y)*((1+Y/
    2)^J-(1+Y/2)^(-C1))-100*(R/2)*J
790 GOTO 850
800 B=(V1+R/2)/(1+Y/2*C1)-R/2*J
```

PROGRAM LISTING

```
 810 !
 820 !
 830 ! View the data
 840 !
 850 PRINT "Number of coupon periods ";F       -Display results
     NR(C1,2) @ GOSUB 940 @ IF NUM(Q$)=8
      THEN 850
 860 PRINT "Annual coupon rate (%) ";R1        -Display next item. If 'BACK'
     @ GOSUB 940 @ IF NUM(Q$)=8 THEN 850        key pressed, show last item
 870 PRINT "Redemption value ";V1 @ GOSU
     B 940 @ IF NUM(Q$)=8 THEN 860
 880 PRINT "Annual yield ";FNR(Y2,3) @ G
     OSUB 940 @ IF NUM(Q$)=8 THEN GOTO 8
     70
 890 PRINT "Bond price ";FNR(B,2) @ GOSU
     B 940 @ IF NUM(Q$)=8 THEN 880
 900 GOTO 1060
 910 !
 920 ! Monitor the keyboard
 930 !
 940 Q$=FNK$ @ IF NUM(Q$)#8 AND NUM(Q$)#       -Monitor keyboard for 'RTN',
     13 AND NUM(Q$)#142 THEN 940                'BACK' or 'TAB keys
 950 IF NUM(Q$)=142 THEN 1090                  -If 'TAB' pressed, quit
 960 RETURN
 970 !
 980 ! Prepend error message
 990 !  to input prompt
1000 !
1010 DISP "Oops...";
1020 RETURN
1030 !
1040 ! Display options menu
1050 !
1060 DISP CHR$(210);"un again,";CHR$(214       -Display program options
     );"iew again, or ";CHR$(197); @ INP
     UT "nd?";Q$
1070 Q$=UPRC$(Q$)
1080 ON POS('RVE',Q$)+1 GOTO 1060,130,85
     0,1090
1090 DELAY 1 @ DISP "" @ STOP
```

PROGRAM DESCRIPTION

DEPRECIATION CALCULATOR

This program will calculate depreciation schedules for investments using straightline, sum-of-years'-digits, declining balance and ACRS methods. The input data are the life of the investment, the cost and salvage value, and the month of purchase. The program will present the depreciation amounts for any year, or print out an entire schedule. For declining balance and ACRS calculations an automatic switchover to straight line is available. In the case of a calculation for a real estate investment with ACRS the calculation will be made with 175% declining balance and automatic switchover to straight-line. The calculations use the following formulae:

N = asset's useful life expectancy

I = starting book value

S = salvage value

F = declining balance factor (%)

j = period number

D(i) = depreciation expense for first period

D(j) = depreciation expense for period j, j=2,3,...N

R(j) = remaining depreciable value at end of period j

M = month of purchase

Yi = 13-M

Straightline:

D(i) = (I-S)/N*Y1/12

D(j) = (I-S)/N

D(N+1) = R(N)

PROGRAM DESCRIPTION

DEPRECIATION CALCULATOR (continued)

Sum-of-years-digits:

D(1) = SOYD(1)*Y1/12

D(j) = SOYD(j)*Y1/12+D(j-1)-d(j-1)*Y1/12

D(N+1) = R(N)

where: SOYD(k) = (N+1-k)/(N*((N+1)/2))*(I-S)

Declining balance:

D(1) = I*F/100*N*Y1/12

D(j) = R(j-1)*F/100*N for j=2,3,...N

D(N+1) = R(N)-S

Accelerated Cost Recovery System:

Lives of assets are recovered over 3, 5, 10, or 15 year periods under the 1981 tax law. The depreciation expense is calculated for each year from a data table. The tables for 1981 are:

Life	Recovery percentage
3	25,38,37
5	15,22,21,21,21
10	8,14,12,10,10,10,9,9,9,9
15	5,10,9,8,7,7,6,6,6,6,6,6,6,6,6

Thus depreciation expense for an asset in year 3 of the 10 year schedule would be I*12/100.

For a real estate asset under ACRS 175% declining balance depreciation is taken with a 15 year life and automatic switchover to straight line.

SAMPLE PROBLEM

Waincorp purchased a computer for $79,500 in March. It has a salvage value of $6,000 and an expected useful life of 8 years. Using 200% declining balance with automatic switch-over, what are the depreciation expenses for years 1, 2, and 3?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	$ Depreciation Calculator $	
2	Select declining balance	Select: SL, SOYD, DB, or ACRS?	DB [RTN]
3	Elect switchover	Switchover to SL (Y/N)?	Y [RTN]
4	Enter cost of investment	Enter cost of investment?	79500 [RTN]
5	Enter useful life	Enter life expectancy?	8 [RTN]
6	Enter month of purchase	Month of purchase (April=4)?	3 [RTN]
7	Enter depreciation factor	Depreciation factor?	200 [RTN]
8	Do not print schedule	Print schedule (Y/N)?	N [RTN]
9	Year 1	Enter year #?	1 [RTN]
		Depreciation = 16562.5	[RTN]
10	Year 2	Enter year #?	2 [RTN]
		Depreciation = 15734.38	[RTN]
11	Year 3	Enter year #?	3 [RTN]
		Depreciation = 11800.78	[RTN]
12	Halt inquiry	Enter year #?	0 [RTN]
13	End program	Run again or End? R	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	$ Depreciation Calculator $	
2	Select depreciation method	Select: SL, SOYD, DB, or ACRS?	
	For straight line, goto step 3		SL [RTN]
	For sum-of-year digit, goto step 11		SOYD [RTN]
	For declining balance, goto step 19		DB [RTN]
	For ACRS goto step 27		ACRS [RTN]
3	Enter cost of investment	Enter cost of investment?	I [RTN]
4	Enter salvage value	Enter salvage value?	S [RTN]
5	Enter useful life	Enter life expectancy?	N [RTN]
6	Enter month # of purchase	Month of purchase (April=4)?	M [RTN]
7	View first year's depreciation	SL Depr 1st year =	[RTN]
8	View subsequent depreciation expense	Straightline =	[RTN]
9	View last year's depreciation	Last year =	[RTN]
10	Goto step 33		
11	Enter cost of investment	Enter cost of investment?	I [RTN]
12	Enter salvage value	Enter salvage value?	S [RTN]
13	Enter useful life	Enter life expectancy?	N [RTN]
14	Enter month # of purchase	Month of purchase (April=4)?	M [RTN]
15	Select print option	Print schedule (Y/N)?	Y or N [RTN]
16	If you select 'Y' the schedule		
	will be printed. Goto step 33		
17	Enter the year # for calculation	Enter year #?	Y [RTN]
	To quit, enter 0 and goto step 33		
18	View depreciation and		
	goto step 17	Depreciation =	[RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
19	Enter switchover preference	Switchover to SL (Y/N)? Y	Y or N [RTN]
20	Enter cost of investment	Enter cost of investment?	I [RTN]
21	Enter useful life	Enter life expectancy?	N [RTN]
22	Enter month # of purchase	Month of purchase (April=4)?	N [RTN]
23	Enter depreciation factor as a		
	percentage, e.g.: 125,200	Depreciation factor?	F [RTN]
24	Select print option. If you		
	select 'Y' then schedule will		
	be printed and goto step 33	Print schedule (Y/N)?	Y or N [RTN]
25	Enter the year # for the cal-		
	culation. Enter 0 to quit and	Enter year #?	Y [RTN]
	goto step 33.		
26	View deprec. & goto step 25	Depreciation	[RTN]
27	Select real estate option. If	Real Estate (Y/N)?	Y or N [RTN]
	you entered 'N' goto step 30		
28	Enter cost of investment	Enter cost of investment?	I [RTN]
29	Goto step 24		
30	Enter cost of investment	Enter cost of investment?	I [RTN]
31	Enter useful life 3,5,10 or 15 years	Enter life expectancy?	N [RTN]
32	Select print option. If you		
	select 'Y' then schedule will		
	be printed and goto step 33.		
	Else goto step 25.	Print schedule (Y/N)?	Y or N [RTN]
33	To run again, enter 'R', to		
	end program enter 'E'.	Run again, or End? R	R or E [RTN]

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
D$	Schedule type	M	Month of purchase
R1	Real estate flag	N	Life expectancy
R2	SOYD remaining balance	X	Loop counter
P1	Print flag	Y	Year #
R	Remaining balance for DB	F	Depreciation factor
I	Investment cost	D1	Depreciation expense
S	Salvage value		

NOTES AND REFERENCES

References: 1. Calculations refer to HP-12C Owner's Handbook.

2. "AN ANALYSIS: 1981 TAX LEGISLATION", Coopers & Lybrand, 1981.

3. Weston & Brigham, MANAGERIAL FINANCE, (The Drylen Press, 1981), p. 60-63.

PROGRAM LISTING

```
 10 ! Depr - Depreciation
 20 ! calculator
 30 !
 40 ! Revision 11/01/82
 50 DEF FNA(X) = INT(X*100+.5)/100           -Round X to two places
 60 DEF FNB                                  -Function to wait for 'RTN' key
 70 IF NUM(KEY$)#13 THEN 70
 80 FNB=0 @ END DEF
 90 DISP '  $ Depreciation Calculator $      -Display sign-on message
    ' @ WAIT 1
100 INPUT 'Select: SL, SOYD, DB, or ACR      -Select depreciation schedule
    S?'; D$ @ D$=UPRC$(D$)
110 IF D$='SL' THEN 160
120 IF D$='SOYD' THEN 210
130 IF D$='DB' THEN 310
140 IF D$='ACRS' THEN 510
150 GOTO 100
160 R1=1 @ GOSUB 640                         -Straight line
170 DISP 'SL Depr 1st year =';FNA((13-M
    )/12*(I-S)/N) @ Z=FNB
180 DISP 'Straightline =';FNA((I-S)/N)
    @ Z=FNB
190 DISP 'Last year =';FNA((I-S)/N-(13-
    M)/12*(I-S)/N) @ Z=FNB
200 GOTO 780
210 R1=1 @ GOSUB 640 @ GOSUB 730 @ IF P      -Sum-of-years digits
    1 THEN Y=N @ GOTO 230
220 GOSUB 750
230 R2=I-S @ FOR X=1 TO Y @ S2=(N+1-X)/
    (N*((N+1)/2))*(I-S)
240 IF X=1 THEN D1=S2*(13-M)/12 @ S1=S2
    -D1
250 IF X>1 THEN D1=S2*(13-M)/12+S1 @ S1
    =S2-S2*(13-M)/12
260 IF P1 THEN PRINT 'Year';X;' =';D1
270 R2=R2-D1 @ NEXT X
280 IF P1 THEN PRINT 'Year';X;' =';R2
290 IF NOT P1 THEN DISP 'Depreciation =
    ';D1 @ Z=FNB @ GOTO 220
300 GOTO 780
310 INPUT 'Switchover to SL (Y/N)','Y';      -Declining balance
     S$ @ IF UPRC$(S$[1,1])='Y' THEN S1
    =1 ELSE S1=0
320 R1=0 @ GOSUB 640
330 INPUT 'Depreciation factor?';F           -Enter depreciation factor
340 GOSUB 730 @ IF NOT P1 THEN 360           -Select print option
350 Y=N @ GOTO 370
360 GOSUB 750
370 R=I @ FOR X=1 TO MIN(Y,N)
380 T1=R*F/N/100 @ IF S1 THEN D1=MAX(T1
    ,(R-S*R1)/(N-X+2-(13-M)/12)) ELSE D
    1=T1
390 IF X=1 THEN D1=(13-M)/12*T1 @ R=R-D
    1 @ GOTO 410
```

PROGRAM LISTING

```
400 R=R-D1                                       -Subtract depreciation from
                                                  remaining book value
410 IF X#Y AND NOT P1 THEN 460
420 IF NOT P1 THEN 450
430 PRINT 'Year';X;' =';FNA(D1)
440 GOTO 460
450 DISP 'Depreciation =';FNA(D1) @ Z=F
    NB
460 NEXT X
470 IF NOT P1 AND Y=N+1 THEN DISP 'Depr
    eciation =';FNA(R-S) @ Z=FNB
480 IF P1 THEN PRINT 'Year';Y+1;' =';FN
    A(R-S)
490 IF NOT P1 THEN GOTO 360
500 GOTO 780
510 INPUT 'Real Estate (Y/N) ?'; R$ @ I     -ACRS depreciation schedule-
    F UPRC$(R$[1,1])='Y' THEN R1=0 @ M=      select real estate option
    6 ELSE R1=1
520 IF NOT R1 THEN S1=1 @ F=175 @ GOSUB     -If real estate use 175%
     640 @ GOTO 340                              declining balance
530 GOSUB 640 @ GOSUB 730                   -Get input data, print option
540 IF N=3 THEN RESTORE 820                 -Select proper table
550 IF N=5 THEN RESTORE 830
560 IF N=10 THEN RESTORE 840
570 IF N=15 THEN RESTORE 850
580 IF P1 THEN 620
590 GOSUB 750 @ FOR X=1 TO Y @ READ D1      -Get year number to display and
    @ NEXT X                                 table value
600 DISP 'Depreciation =';I*D1/100          -Display result
610 Z=FNB @ GOTO 540                        -Wait for return key and ask
                                             for year number again
620 FOR X=1 TO N @ READ D1
630 PRINT 'Year';X;' =';I*D1/100 @ NEXT
     X @ GOTO 780
640 INPUT 'Enter cost of investment?';I     -Accept input data
650 S=0 @ IF D$#'ACRS' AND D$#'DB' THEN     -Get salvage value for SL and
     INPUT 'Enter salvage value?';S          SOYD
660 IF NOT R1 AND D$#'DB' THEN N=15 @ R     -Set 15 year life for ACRS real
    ETURN                                    estate option
670 INPUT 'Enter life expectancy?';N
680 IF D$='ACRS' THEN 720
690 INPUT 'Month of purchase (April=4)?
    ';M
700 IF M<1 OR M>12 THEN BEEP @ DISP 'In
    valid month' @ GOTO 690
710 RETURN
720 IF N#3 AND N#5 AND N#10 AND N#15 TH     -Check validity of life for
    EN BEEP @ DISP 'Invalid life' @ GOT      ACRS
    O 670 ELSE RETURN
730 INPUT 'Print schedule (Y/N) ?'; P$      -Select print option
    @ IF UPRC$(P$[1,1])='Y' THEN P1=1 E
    LSE P1=0
740 RETURN
750 INPUT 'Enter year #?';Y                 -Get year number
```

PROGRAM LISTING

```
760 Y=INT(ABS(Y)) @ IF Y>0 AND Y<=N+1 T
    HEN RETURN
770 IF Y#0 THEN BEEP @ DISP 'Year out o
    f range!' @ GOTO 750
780 DISP CHR$(210);'un again or ';CHR$(     -Program options
    197);
790 INPUT 'nd?','R'; Q$ @ Q$=UPRC$(Q$[1
    ,1])
800 ON POS('RE',Q$)+1 GOTO 780,100,810
810 DISP @ STOP
820 DATA 25,38,37                           -1981 ACRS tables - three year
                                             table
830 DATA 15,22,21,21,21                     -5 year table
840 DATA 8,14,12,10,10,10,9,9,9,9           -10 year table
850 DATA 5,10,9,8,7,7,6,6,6,6,6,6,6,6,6     -15 year table
```

PROGRAM DESCRIPTION

LEASE VERSUS PURCHASE

An investment decision frequently encountered is the decision to lease or purchase capital equipment. Although a thorough evaluation of a complex acquisition usually requires the services of a qualified accountant, it is possible to simplify a number of assumptions and use annual cash flow estimates to produce a first approximation.

The program assumes that the purchase is financed with a loan and that the loan is made for the term of the lease. (The term may be either 3, 5, 10, or 15 years, to correspond with the ACRS depreciation schedule). The tax advantages of interest paid, depreciation, and the investment credit which accrues from ownership are compared to the tax advantage of treating the lease payment as an expense. The resulting cash flows are discounted to the present at the firm's after-tax cost of capital.

The program displays the net advantage of owning vs. leasing for each year of the analysis as well as reporting the total net advantage at the end of all the years.

A negative value for the net advantage indicates that a lease is a better choice.

SAMPLE PROBLEM

Home Style Bagel Co. is evaluating the acquisition of a mixer which can be leased for $1,700 per year with the first and last payments in advance and a $750 buy-back option at the end of 10 years. The lease includes maintenance. The same equipment could be purchased for $10,000 with a 12% loan. Maintenance is assumed to be 2% of the purchase price for the first four years. A major overhaul is predicted for the fifth year at a cost of $1,500. Subsequent yearly maintenance of 3% is estimated for the remainder of the 10 year term. The company would use the ACRS method of depreciation on a 10 year life with no salvage value. An accountant informs management to take the 10% capital investment tax credit at the end of the second year and to figure the cash flows at a 48% tax rate. The after-tax cost of capital (discounting rate) is 5%.

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	Sign-on message	$ Lease vs Purchase $	
2	Enter data	Life of investment?	10 [RTN]
		Principal of loan?	10000 [RTN]
		Loan interest rate (%)?	12 [RTN]
		Marginal tax rate (%)?	48 [RTN]
		Discount rate (%)?	5 [RTN]
		Initial year; lease amt?	3400 [RTN]
		Year: 1	
		Lease payment amount?	1700 [RTN]
		Maintenance expenses?	200 [RTN]
		Net advantage is: 1739.58	[RTN]
		Year: 2	
		Lease payment amount?	1700 [RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Maintenance expenses?	200 [RTN]
		Net advantage is: 1943.97	[RTN]
		Year: 3	
		Lease payment amount?	1700 [RTN]
		Maintenance expenses?	200 [RTN]
		Net advantage is 2023.93	[RTN]
		Year: 4	
		Lease payment amount?	1700 [RTN]
		Maintenance expenses	200 [RTN]
		Net advantage is 1987.24	[RTN]
		Year: 5	
		Lease payment amount?	1700 [RTN]
		Maintenance expenses?	1500 [RTN]
		Net advantage is 1386.5	[RTN]
		Year: 6	
		Lease payment amount?	1700 [RTN]
		Maintenance expenses?	300 [RTN]
		Net advantage is 1028.12	[RTN]
		Year: 7	
		Lease payment amount?	1700 [RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Maintenance expenses?	300 [RTN]
		Net advantage is: 1028.12	[RTN]
		Year: 8	
		Lease payment amount?	1700 [RTN]
		Maintenance expenses?	300 [RTN]
		Net advantage is 781.08	[RTN]
		Year: 9	
		Lease payment amount?	0 [RTN]
		Maintenance expenses?	300 [RTN]
		Net advantage is: -70.81	[RTN]
		Year: 10	
		Lease payment amount?	0 [RTN]
		Maintenance expenses?	300 [RTN]
		Net advantage is: -932.02	[RTN]
		Amt and year of tax credit:	1000,2 [RTN]
		Amount of buy-back	750 [RTN]
3	Display result	Final net advantage: 214.44	[RTN]
4	End program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	See sign-on message :	$ Lease vs Purchase $	
2	Enter data :	Life of investment?	N [RTN]
		Principal of loan?	P [RTN]
	Entering "Q" at any time will	Loan interest rate (%)?	I [RTN]
	cause the program to advance	Marginal tax rate (%)?	T [RTN]
	to step 4.	Discount rate (%)?	D [RTN]
		Initial year: lease amt?	L [RTN]
		Year i	
	Perform these steps for all	Lease payment amount?	L [RTN]
	years of analysis using annual	Maintenance expenses?	M [RTN]
	data.	Net advantage is N2	[RTN]
	Enter tax credit and buy-back :	Amt and year of tax credit	C,Z1 [RTN]
		Amount of buy-back:	B1 [RTN]
3	Display results :	Final net advantage: $ N2	[RTN]
4	Display options menu :	Run again, View again, or End?	R,V,or E [RTN]
	If 'R' pressed goto step 1a		
	If 'E' pressed then stop		
	If 'V' pressed goto step 5		
5	View the data :	Life of investment	[RTN]
	[RTN] advances to next item	Principal of loan	[RTN]
	[BACK] shows prior item	Loan interest rate (%)	[RTN]
	[TAB] goes to step 4	Marginal tax rate (%)	[RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Discount rate (%)	[RTN]
		Final net advantage:	[RTN]
6	Goto step 4		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A	ACRS depreciation rate	L	Amount of lease payment in year Z
A1( )	Dollar amount to interest	M	Maintenance expense in year Z
A2	Amount amortized	N	Life of investment
B	Balance due on loan (starts = P)	N2	Net advantage
B1	Dollar amount of buy-back in year N	P	Loan principal
B9	Net present value (NPV) of buy-back	R1	Annual payment on loan
C	Tax credit in year Z1	T	Marginal tax rate (percent)
C0	Cost of owning	T1	Marginal tax rate (decimal)
C1	Cost of leasing	T8	Total amount amortized
C9	NPV of tax credit	X	Target number in rounding function
D	Discount rate (percentage)	Z	Year in life of investment
D2	Discount rate (decimal)	Z1	Year of tax credit
D9( )	Dollar amount of ACRS depreciation	K$	Key pressed by user
I	Loan interest rate (percentage)	Q$	Key pressed by user
I1	Loan interest rate (decimal)	X$	Alpha variables used to input data
J	Precision in rounding function	Y$	Usually converted to numeric

NOTES AND REFERENCES

Note: The computations assume ANNUAL data only.

Reference: HP-12C Solutions Handbook, pp. 49, 143.

PROGRAM LISTING

```
10 ! Given the principal, life,
20 ! and interest rate of a loan;
30 ! given the lease payments,
40 ! discount rate, maintenance,
50 ! tax credit, and buy-back,
60 ! compute the net advantage of
70 ! leasing capital equipment.
80 !
90 ! Revision 11/01/82
100 !
110 DIM D9(15),A1(15)
120 !
130 ! Round X to J decimal places
140 !
150 DEF FNR(X,J) = INT(X*10^J+.5)/10^J
160 !
170 ! Single upper-case key in
180 !
190 DEF FNK$
200 K$=KEY$ @ IF K$="" THEN 200
210 FNK$=UPRC$(K$)
220 END DEF
230 DISP "      $ Lease vs Purchase $"
240 !
250 ! Begin to gather the data
260 !
270 INPUT "Life of investment? "; X$ @      -Accept input, set error trap
    ON ERROR GOTO 310
280 IF X$="" THEN 270
290 IF X$="Q" THEN 1500                     -If 'Q' then goto program
                                             options
300 N=VAL(X$) @ OFF ERROR @ GOTO 320        -Convert to numeric
310 DISP "Oops..."; @ GOTO 270              -Prepend error message to
                                             prompt and ask again
320 IF N#3 AND N#5 AND N#10 AND N#15 TH     -Check life of investment
    EN BEEP @ GOTO 310
330 INPUT "Principal of Loan? "; X$ @ O
    N ERROR GOTO 370
340 IF X$="" THEN GOTO 330
350 IF X$="Q" THEN GOTO 1500
360 P=VAL(X$) @ OFF ERROR @ GOTO 380
370 DISP "Oops..."; @ GOTO 330
380 INPUT "Loan interest rate? (%)"; X$
     @ ON ERROR GOTO 420
390 IF X$="" THEN GOTO 380
400 IF X$="Q" THEN GOTO 1500
410 I=VAL(X$) @ OFF ERROR @ I1=I/100 @
    GOTO 430
420 DISP "Oops..."; @ GOTO 380
430 INPUT "Marginal tax rate? (%)"; X$
    @ ON ERROR GOTO 470
440 IF X$="" THEN GOTO 430
450 IF X$="Q" THEN GOTO 1500
460 T=VAL(X$) @ OFF ERROR @ T1=T/100 @
    GOTO 480
```

PROGRAM LISTING

```
470 DISP "Oops..."; @ GOTO 430
480 INPUT "Discount rate (%)? "; X$ @ O
    N ERROR GOTO 520
490 IF X$="" THEN 480
500 IF X$="Q" THEN 1500
510 D=VAL(X$) @ OFF ERROR @ D2=D/100 @
    GOTO 560
520 DISP "Oops..."; @ GOTO 480
530 !
540 ! Select depr. schedule
550 !
560 IF N=3 THEN RESTORE 1540              -Select depr. schedule
570 IF N=5 THEN RESTORE 1550
580 IF N=10 THEN RESTORE 1560
590 IF N=15 THEN RESTORE 1570
600 !
610 ! Compute annual depr.
620 !
630 FOR X=1 TO N @ READ A                 -Build array of depreciation
                                           amounts
640 D9(X)=P*A/100 @ NEXT X
650 !
660 ! Compute the annual pmt
670 !
680 R1=I1*P/(1-(1+I1)^(-N))               -Compute annual payment
690 B=P @ T8=0
700 INPUT "Initial year: lease amt? ";    -Get initial lease amount
    X$ @ ON ERROR GOTO 740
710 IF X$="" THEN 700
720 IF X$="Q" THEN 1500
730 L=VAL(X$) @ OFF ERROR @ GOTO 750
740 DISP "Oops..."; @ GOTO 700
750 N2=(1-T1)*L
760 DISP
770 !
780 ! Amortize loan
790 !
800 FOR Z=1 TO N @ A1(Z)=B*I1 @ A2=R1-A   -Build array of interest
    1(Z)                                   payments
810 T8=T8+A2 @ B=B-A2 @ NEXT Z
820 !
830 ! Display current year
840 !
850 FOR Z=1 TO N @ DISP "    Year: ";Z    -Accept lease payment, maint.
                                           for each year and compute
860 INPUT "Lease payment amount? "; X$
    @ ON ERROR GOTO 900
870 IF X$="" THEN GOTO 860
880 IF X$="Q" THEN GOTO 1500
890 L=VAL(X$) @ OFF ERROR @ GOTO 910
900 DISP "Oops..."; @ GOTO 860
910 INPUT "Maintenance expenses? "; X$
    @ ON ERROR GOTO 950
920 IF X$="" THEN 910
```

PROGRAM LISTING

```
930 IF X$="Q" THEN 1500
940 M=VAL(X$) @ OFF ERROR @ GOTO 990
950 DISP "Oops...";  @ GOTO 910
960 !
970 ! Compute:
980 !
990 C1=(1-T1)*L
1000 C0=R1-T1*(A1(Z)+D9(Z))+(1-T1)*M
1010 N2=N2+(C1-C0)/(1+D2)^Z
1020 DISP "Net advantage is: ";FNR(N2,2)
     @ GOSUB 1440 @ IF NUM(Q$)=8 THEN 1
     020
1030 DISP @ DISP
1040 NEXT Z
1050 !
1060 ! Adjustments for tax
1070 ! credits and buy-backs
1080 !
1090 INPUT "Amt and year of tax credit:      -Adjustments for investment tax
     "; X$,Y$ @ ON ERROR GOTO 1130           credit and buy-back
1100 IF X$="" OR Y$="" THEN 1090
1110 IF X$="Q" OR Y$="Q" THEN 1500
1120 C=VAL(X$) @ Z1=VAL(Y$) @ OFF ERROR
     @ GOTO 1180
1130 DISP "Oops..."; @ GOTO 1090
1140 !
1150 ! Compute the net present
1160 ! value of the tax credit
1170 !
1180 C9=C/(1+D2)^Z1
1190 N2=N2+C9
1200 INPUT "Amount of buy-back: "; X$ @
     ON ERROR GOTO 1240
1210 IF X$="" THEN 1200
1220 IF X$="Q" THEN 1500
1230 B=VAL(X$) @ OFF ERROR @ GOTO 1290
1240 DISP "Oops..."; @ GOTO 1200
1250 !
1260 ! Compute net present value
1270 ! of the buy-back
1280 !
1290 B9=B*(1-T1)/(1+D2)^N
1300 N2=N2+B9
1310 DISP "Final net advantage: $";FNR(N    -Display final net advantage
     2,2) @ GOSUB 1440 @ IF NUM(Q$)=8 TH
     EN 1310 ELSE 1500
1320 !
1330 ! View data and results
1340 !
1350 DISP "Life of investment: ";N @ GOS
     UB 1440 @ IF NUM(Q$)=8 THEN 1350
1360 DISP "Principal of loan: ";P @ GOSU
     B 1440 @ IF NUM(Q$)=8 THEN 1350
1370 DISP "Loan interest rate ";I;"%" @
     GOSUB 1440 @ IF NUM(Q$)=8 THEN GOTO
      1360
```

PROGRAM LISTING

```
1380 DISP "Marginal tax rate ";T;"%" @ G
     OSUB 1440 @ IF NUM(Q$)=8 THEN GOTO
     1370
1390 DISP "Discount rate ";D;"%" @ GOSUB
      1440 @ IF NUM(Q$)=8 THEN GOTO 1380
1400 DISP "Final net advantage: $";FNR(N
     2,2) @ GOSUB 1440 @ IF NUM(Q$)=8 TH
     EN 1390 ELSE 1500
1410 !
1420 ! Monitor the keyboard
1430 !
1440 Q$=FNK$ @ IF NUM(Q$)#8 AND NUM(Q$)#
     13 AND NUM(Q$)#142 THEN 1440
1450 IF NUM(Q$)=142 THEN 1500
1460 RETURN
1470 !
1480 ! Display options menu
1490 !
1500 DISP CHR$(210);"un again,";CHR$(214
     );"iew again, or ";CHR$(197); @ INP
     UT "nd ";Q$
1510 Q$=UPRC$(Q$)
1520 ON POS('RVE',Q$)+1 GOTO 1500,230,13
     50,1530
1530 DISP "" @ STOP
1540 DATA 25,38,37                          -1981 ACRS tax tables
1550 DATA 15,22,21,21,21
1560 DATA 8,14,12,10,10,10,9,9,9,9
1570 DATA 5,10,9,8,7,7,6,6,6,6,6,6,6,6,6
```

PROGRAM DESCRIPTION

PRESENT VALUE OF A GEOMETRIC SERIES

This program computes the present value of a series of cash flows that changes over time, such as with inflation. Example required inputs are the payment, growth rate, discount rate, and the number of periods. The period for the payment is the same as the period for the growth and discount rates.

SAMPLE PROBLEM

What sum must a person have in an education fund if they wish to draw from the fund purchasing power equal to $550 per month for five years? Assume a monthly inflation rate of .67% (8% annually), and a discount rate of .56% (6.75% annually).

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run the program		
1a	See sign-on message	$ Geometric Gradient $	
2	Enter the data	What is the payment?	550 [RTN]
		How many periods?	60 [RTN]
		Discount rate?	.56 [RTN]
		Growth rate?	.67 [RTN]
3	Display results	PV = 33897.93	[RTN]
4	Program options	Run again, View again, or End?	V [RTN]
5	View data and results	Payment = 550	[RTN]
		Discount rate = .56	[RTN]
		Growth rate = .67	[RTN]
		Periods = 60	[RTN]
		Present value = 33897.93	[RTN]
6	End program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	See sign-on message	$ Geometric Gradient $	
2	Enter data as requested:	What is payment?	P [RTN]
	An uppercase "Q" entered at	How many periods?	N [RTN]
	this time will cause the	Discount rate?	I [RTN]
	program to goto step 4.	Growth rate?	G [RTN]
3	Display results.	PV =	[RTN]
4	Display options menu:	Run again, View again, or End?	R,V,or E [RTN]
	If 'R' then goto 1a		
	If 'E' then program stops.		
5	If 'V' then view data and	Payment =	[RTN]
	results. [RTN] advances to	Discount rate =	[RTN]/[BACK]
	next item. [BACK] shows	Growth rate =	[RTN]/[BACK]
	previous entry. [TAB] goes	Periods =	[RTN]/[BACK]
	to step 4.	Present value =	[RTN]/[BACK]
	Goto step 4.		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
G	Growth rate as a percentage	P	Payment in series
G1	Growth rate as a decimal fraction	V	Present value of series
I	Discount rate as a percentage	X	Target value in rounding function FNR
I1	Discount rate as a decimal fraction	K$	Key pressed by user
J	Precision in rounding fraction FNR	Q$	Keyboard response
K	Temporary variable	X$	Alpha input value, converted to decimal by routine
N	Number of time periods		

NOTES AND REFERENCES

Notes:

1. The discount rate is assumed to be an annual value. If the data are not annual, the discount rate has to be adjusted appropriately.

2. The growth rate is assumed to be an annual value. If the data are not annual, the growth rate has to be adjusted appropriately.

3. The payments are assumed to be end of the period payments.

Reference:

Stermole, F.J., ECONOMIC EVALUATION AND INVESTMENT DECISION METHODS, Investment Evaluations Corp., 1974, Appendices E and F.

PROGRAM LISTING

```
 10 ! compute the present value
 20 ! of a geometric gradient
 30 ! series for a finite number
 40 ! of periods
 50 !
 60 ! Revision 11/01/82
 70 !
 80 DISP "    $ Geometric Gradient $"        -Display sign-on
 90 !
100 ! round X to J decimal places
110 !
120 DEF FNR(X,J) = INT(X*10^J+.5)/10^J
130 !
140 ! monitor keyboard,returning
150 ! key value
160 !
170 DEF FNK$
180 K$=KEY$ @ IF K$="" THEN 180
190 FNK$=UPRC$(K$)
200 END DEF
210 !
220 ! get the input data
230 !
240 DISP "What is the payment"; @ INPUT   -Accept input, set error trap
     X$ @ ON ERROR GOTO 280
250 IF X$="" THEN 240                     -Trap null input
260 IF X$="Q" THEN 780                    -If 'Q' then goto program
                                           options
270 P=VAL(X$) @ OFF ERROR @ GOTO 290      -Convert to numeric
280 DISP "Oops..."; @ GOTO 240            -Prepend error message to
                                           prompt and ask again
290 DISP "How many periods"; @ INPUT X$
     @ ON ERROR GOTO 330
300 IF X$="" THEN 290
310 IF X$="Q" THEN 780
320 N=VAL(X$) @ OFF ERROR @ GOTO 340
330 DISP "Oops..."; @ GOTO 290
340 DISP "Discount rate"; @ INPUT X$ @
    ON ERROR GOTO 380
350 IF X$="" THEN 340
360 IF X$="Q" THEN 780
370 I=VAL(X$) @ OFF ERROR @ I1=I/100 @
    GOTO 390
380 DISP "Oops..."; @ GOTO 340
390 DISP "Growth rate"; @ INPUT X$ @ ON
     ERROR GOTO 430
400 IF X$="" THEN 390
410 IF X$="Q" THEN 780
420 G=VAL(X$) @ OFF ERROR @ G1=G/100 @
    GOTO 440
430 DISP "Oops..."; @ GOTO 390
440 IF I=G THEN V=P*N @ GOTO 510
450 !
460 K=(1+G1)/(1+I1)-1                     -Compute results
```

PROGRAM LISTING

```
470 V=P*(1/(1+I1))*(((1+K)^N-1)/K)
480 !
490 ! display the present value
500 !
510 DISP "PV = ";FNR(V,2) @ GOSUB 720 @        -Display result
     IF NUM(Q$)=8 THEN GOTO 510
520 GOTO 780
530 !
540 ! view input and results
550 !
560 DISP "Payment = ";P @ GOSUB 720            -View input data and result
570 IF NUM(Q$)=8 THEN GOTO 560
580 DISP "Discount rate = ";I @ GOSUB 7
    20
590 IF NUM(Q$)=8 THEN GOTO 560                 -If 'BACK' key pressed show
                                                previous item
600 DISP "Growth rate = ";G @ GOSUB 720
610 IF NUM(Q$)=8 THEN 580
620 DISP "Periods = ";N @ GOSUB 720
630 IF NUM(Q$)=8 THEN GOTO 600
640 DISP "Present value = ";FNR(V,2) @
    GOSUB 720
650 IF NUM(Q$)=8 THEN GOTO 620
660 GOTO 780
670 !
680 ! monitor keyboard and
690 ! accept only RTN, BACK,or
700 ! TAB as valid keys
710 !
720 Q$=FNK$ @ IF NUM(Q$)#13 AND NUM(Q$)
    #8 AND NUM(Q$)#142 THEN 720
730 IF NUM(Q$)=142 THEN 780
740 RETURN
750 !
760 ! display options menu
770 !
780 DISP CHR$(210);"un again, ";CHR$(21        -Display options
    4);"iew again, or ";CHR$(197);
790 INPUT "nd?"; Q$ @ Q$=UPRC$(Q$)
800 ON POS("RVE",Q$)+1 GOTO 780,80,560,        -Accept only 'R', 'V', or 'E'
    810                                         keys
810 DISP @ STOP
```

PROGRAM DESCRIPTION

PRESENT VALUE OF AN ARITHMETIC GRADIENT SERIES

This program is used to compute the present value of a series of cash flows that changes arithmetically. Example required inputs are the initial cash flow, the amount of the payment that changes the cash flows, the number of periods in the series, and the interest rate. Note that the period of the interest rate is the same as the payment period.

SAMPLE PROBLEM

The after-tax expenses on a machine are expected to begin at $1,200 at the end of the first year and increase by $350 at the end of each year over the 12-year life of the machine. What is the present value of the series if it is discounted at 12%?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	See sign-on message	$ Arithmetic Gradient $	
2	Enter the data	What is the 1st cash flow?	1200 [RTN]
		What is the payment?	350 [RTN]
		How many periods?	12 [RTN]
		Discount rate?	12 [RTN]
3	Display answer	PV = 16516.55	[RTN]
4	Present options menu	Run again, View again, or End?	V [RTN]
5	View the data and results	1st cash flow = 1200	[RTN]
		Payment = 350	[RTN]
		Discount rate = 12	[RTN]
		Periods = 12	[RTN]
		Present value = 16516.55	[RTN]
6	Present options menu	Run again, View again, or End?	E [RTN]
	End program		

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
1a	Sign-on message	$ Arithmetic Gradient $	
2	Enter data	What is the 1st cash flow?	S [RTN]
		What is the payment	p [RTN]
	An uppercase "Q" at this time	How many periods?	n [RTN]
	will cause the progarm to go		
	to step 4	Discount rate?	i [RTN]
3	Display results	PV =	[RTN]
4	Program options:	Run again, View again, or End?	R, V, or E [RTN]
	If 'R' then goto 1a		
	If 'E' then end program		
	If 'V' then: 3		
5	View the data and results	1st cash flow =	[RTN]
		Payment =	[RTN]/[BACK]
		Discount rate =	[RTN]/[BACK]
		Periods =	[RTN]/[BACK]
		Present value =	[RTN]/[BACK]
	Goto step 4		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
I	Discount rate as a percentage	W	Temporary value
I1	Discount rate as a decimal function	X	Number to be rounded in FNR
J	Precision in rounding function FNR	K$	Key pressed
N	Number of periods in analysis	Q$	Alpha input; used to control the program execution
P	Payment		
V	Present value of payment over N time periods at I discount rate	X$	Input value; converted to numeric
		S	First cash flow

NOTES AND REFERENCES

Reference: Stermole, F.J., ECONOMIC EVALUATION AND INVESTMENT DECISION METHODS, Investment Evaluations Corp., 1974, Appendices E and F.

PROGRAM LISTING

```
 10 ! Compute the present value
 20 ! of an arithmetic gradient
 30 ! series for a finite number
 40 ! of periods
 50 !
 60 !
 70 ! Revision 11/01/82
 80 !
 90 DISP "    $ Arithmetic Gradient $"      -Display sign-on message
100 !
110 ! Round X to J decimal places
120 !
130 DEF FNR(X,J) = INT(X*10^J+.5)/10^J
140 !
150 ! Monitor keyboard, returning
160 ! key value
170 !
180 DEF FNK$
190 K$=KEY$ @ IF K$="" THEN 190
200 FNK$=UPRC$(K$)
210 END DEF
220 !
230 ! get the input data
240 !
250 DISP "What is the 1st cash flow"; @    -Accept input, set error trap
     INPUT X$ @ ON ERROR GOTO 290
260 IF X$="" THEN 250                      -If null input, ask again
270 IF X$='Q' THEN 800                     -If 'Q' entered, goto program
                                            options
280 S=VAL(X$) @ OFF ERROR @ GOTO 300       -Convert to numeric
290 DISP 'Oops...' @ GOTO 250              -Prepend error to prompt and
                                            ask again
300 DISP "What is the payment"; @ INPUT
     X$ @ ON ERROR GOTO 340
310 IF X$="" THEN 300
320 IF X$="Q" THEN 800
330 P=VAL(X$) @ OFF ERROR @ GOTO 350
340 DISP "Oops..."; @ GOTO 300
350 DISP "How many periods"; @ INPUT X$
     @ ON ERROR GOTO 390
360 IF X$="" THEN 350
370 IF X$="Q" THEN 800
380 N=VAL(X$) @ OFF ERROR @ GOTO 400
390 DISP "Oops..."; @ GOTO 350
400 DISP "Discount rate"; @ INPUT X$ @
    ON ERROR GOTO 440
410 IF X$="" THEN 400
420 IF X$="Q" THEN 800
430 I=VAL(X$) @ OFF ERROR @ I1=I/100 @
    GOTO 470
440 DISP "Oops..."; @ GOTO 400
450 !
460 ! compute the values...                  -Compute result
470 W=(1-1/(1+I1)^N)/I1
```

PROGRAM LISTING

```
480 V=W*P*(1/I1-N/(I1*W*(1+I1)^N))
490 V=V+S*W
500 !
510 ! display the present value
520 !
530 DISP "PV = ";FNR(V,2) @ GOSUB 740 @     -Display result
     IF NUM(Q$)=8 THEN GOTO 530
540 GOTO 800
550 !
560 ! view the data and results
570 !
580 DISP "1st cash flow = ";S @ GOSUB 7     -View data and results
    40
590 IF NUM(Q$)=8 THEN GOTO 580
600 DISP "Payment = ";P @ GOSUB 740
610 IF NUM(Q$)=8 THEN GOTO 580              -If 'BACK' key pressed, show
                                             previous item
620 DISP "Discount rate = ";I @ GOSUB 7
    40
630 IF NUM(Q$)=8 THEN GOTO 600
640 DISP "Periods = ";N @ GOSUB 740
650 IF NUM(Q$)=8 THEN GOTO 620
660 DISP "Present value = ";FNR(V,2) @
    GOSUB 740
670 IF NUM(Q$)=8 THEN GOTO 640
680 GOTO 800
690 !
700 ! monitor keyboard and
710 ! accept only RTN, BACK, or
720 ! TAB as valid keys
730 !
740 Q$=FNK$ @ IF NUM(Q$)#13 AND NUM(Q$)
    #8 AND NUM(Q$)#142 THEN 740
750 IF NUM(Q$)=142 THEN 800
760 RETURN
770 !
780 ! display options menu
790 !
800 DISP CHR$(210);"un again, ";CHR$(21
    4);"iew again, or ";CHR$(197);
810 INPUT "nd?"; Q$ @ Q$=UPRC$(Q$)
820 ON POS("RVE",Q$)+1 GOTO 800,90,580,     -Accept only 'R', 'V' or 'E'
    830
830 DISP @ STOP
```

NOTES

NOTES

NOTES

FINANCE

ALL HP-75 SOLUTIONS BOOKS ARE AVAILABLE RECORDED ON MINI-DATA CASSETTES FROM EITHER A HEWLETT-PACKARD DEALER OR THE HP USERS' LIBRARY.

HEWLETT
PACKARD

00075-90071 Printed in U.S.A.

Reorder No.
00075-13009